NAG Library Routine Document

F07FSF (ZPOTRS)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F07FSF (ZPOTRS) solves a complex Hermitian positive definite system of linear equations with multiple right-hand sides,

$$AX = B,$$

where A has been factorized by F07FRF (ZPOTRF).

2 Specification

```
SUBROUTINE F07FSF (UPLO, N, NRHS, A, LDA, B, LDB, INFO)
INTEGER                N, NRHS, LDA, LDB, INFO
COMPLEX (KIND=nag_wp) A(LDA,*), B(LDB,*)
CHARACTER(1)           UPLO
```

The routine may be called by its LAPACK name ***zpotrs***.

3 Description

F07FSF (ZPOTRS) is used to solve a complex Hermitian positive definite system of linear equations $AX = B$, this routine must be preceded by a call to F07FRF (ZPOTRF) which computes the Cholesky factorization of A. The solution X is computed by forward and backward substitution.

If UPLO = 'U', $A = U^{\mathrm{H}}U$, where U is upper triangular; the solution X is computed by solving $U^{\mathrm{H}}Y = B$ and then $UX = Y$.

If UPLO = 'L', $A = LL^{\mathrm{H}}$, where L is lower triangular; the solution X is computed by solving $LY = B$ and then $L^{\mathrm{H}}X = Y$.

4 References

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Parameters

1: UPLO – CHARACTER(1) *Input*

On entry: specifies how A has been factorized.

UPLO = 'U'
$A = U^{\mathrm{H}}U$, where U is upper triangular.

UPLO = 'L'
$A = LL^{\mathrm{H}}$, where L is lower triangular.

Constraint: UPLO = 'U' or 'L'.

2: N – INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: N ≥ 0.

3: NRHS – INTEGER *Input*

On entry: r, the number of right-hand sides.

Constraint: $\text{NRHS} \geq 0$.

4: $\text{A}(\text{LDA}, *)$ – COMPLEX (KIND=nag_wp) array *Input*

Note: the second dimension of the array A must be at least $\max(1, \text{N})$.

On entry: the Cholesky factor of A, as returned by F07FRF (ZPOTRF).

5: LDA – INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F07FSF (ZPOTRS) is called.

Constraint: $\text{LDA} \geq \max(1, \text{N})$.

6: $\text{B}(\text{LDB}, *)$ – COMPLEX (KIND=nag_wp) array *Input/Output*

Note: the second dimension of the array B must be at least $\max(1, \text{NRHS})$.

On entry: the n by r right-hand side matrix B.

On exit: the n by r solution matrix X.

7: LDB – INTEGER *Input*

On entry: the first dimension of the array B as declared in the (sub)program from which F07FSF (ZPOTRS) is called.

Constraint: $\text{LDB} \geq \max(1, \text{N})$.

8: INFO – INTEGER *Output*

On exit: $\text{INFO} = 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

$\text{INFO} < 0$

If $\text{INFO} = -i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

For each right-hand side vector b, the computed solution x is the exact solution of a perturbed system of equations $(A + E)x = b$, where

if UPLO = 'U', $|E| \leq c(n)\epsilon|U^{\mathrm{H}}||U|$;

if UPLO = 'L', $|E| \leq c(n)\epsilon|L||L^{\mathrm{H}}|$,

$c(n)$ is a modest linear function of n, and ϵ is the ***machine precision***.

If $\hat{x}$ is the true solution, then the computed solution x satisfies a forward error bound of the form

$$\frac{\|x - \hat{x}\|_\infty}{\|x\|_\infty} \leq c(n)\,\text{cond}(A, x)\epsilon$$

where $\text{cond}(A, x) = \left\||A^{-1}||A||x|\right\|_\infty / \|x\|_\infty \leq \text{cond}(A) = \left\||A^{-1}||A|\right\|_\infty \leq \kappa_\infty(A)$.

Note that $\text{cond}(A, x)$ can be much smaller than $\text{cond}(A)$.

Forward and backward error bounds can be computed by calling F07FVF (ZPORFS), and an estimate for $\kappa_\infty(A)$ $(= \kappa_1(A))$ can be obtained by calling F07FUF (ZPOCON).

8 Parallelism and Performance

F07FSF (ZPOTRS) is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

F07FSF (ZPOTRS) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

The total number of real floating-point operations is approximately $8n^2r$.

This routine may be followed by a call to F07FVF (ZPORFS) to refine the solution and return an error estimate.

The real analogue of this routine is F07FEF (DPOTRS).

10 Example

This example solves the system of equations $AX = B$, where

$$A = \begin{pmatrix} 3.23 + 0.00i & 1.51 - 1.92i & 1.90 + 0.84i & 0.42 + 2.50i \\ 1.51 + 1.92i & 3.58 + 0.00i & -0.23 + 1.11i & -1.18 + 1.37i \\ 1.90 - 0.84i & -0.23 - 1.11i & 4.09 + 0.00i & 2.33 - 0.14i \\ 0.42 - 2.50i & -1.18 - 1.37i & 2.33 + 0.14i & 4.29 + 0.00i \end{pmatrix}$$

and

$$B = \begin{pmatrix} 3.93 - 6.14i & 1.48 + 6.58i \\ 6.17 + 9.42i & 4.65 - 4.75i \\ -7.17 - 21.83i & -4.91 + 2.29i \\ 1.99 - 14.38i & 7.64 - 10.79i \end{pmatrix}.$$

Here A is Hermitian positive definite and must first be factorized by F07FRF (ZPOTRF).

10.1 Program Text

```
    Program f07fsfe

!     F07FSF Example Program Text

!     Mark 25 Release. NAG Copyright 2014.

!     .. Use Statements ..
      Use nag_library, Only: nag_wp, x04dbf, zpotrf, zpotrs
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Integer                          :: i, ifail, info, lda, ldb, n, nrhs
      Character (1)                    :: uplo
!     .. Local Arrays ..
      Complex (Kind=nag_wp), Allocatable :: a(:,:), b(:,:)
      Character (1)                    :: clabs(1), rlabs(1)
!     .. Executable Statements ..
      Write (nout,*) 'F07FSF Example Program Results'
!     Skip heading in data file
      Read (nin,*)
      Read (nin,*) n, nrhs
      lda = n
```

```
      ldb = n
      Allocate (a(lda,n),b(ldb,nrhs))

!     Read A and B from data file

      Read (nin,*) uplo
      If (uplo=='U') Then
        Read (nin,*)(a(i,i:n),i=1,n)
      Else If (uplo=='L') Then
        Read (nin,*)(a(i,1:i),i=1,n)
      End If
      Read (nin,*)(b(i,1:nrhs),i=1,n)

!     Factorize A
!     The NAG name equivalent of zpotrf is f07frf
      Call zpotrf(uplo,n,a,lda,info)

      Write (nout,*)
      Flush (nout)
      If (info==0) Then

!       Compute solution
!       The NAG name equivalent of zpotrs is f07fsf
        Call zpotrs(uplo,n,nrhs,a,lda,b,ldb,info)

!       Print solution

!       ifail: behaviour on error exit
!              =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
        ifail = 0
        Call x04dbf('General',' ',n,nrhs,b,ldb,'Bracketed','F7.4', &
          'Solution(s)','Integer',rlabs,'Integer',clabs,80,0,ifail)

      Else
        Write (nout,*) 'A is not positive definite'
      End If

    End Program f07fsfe
```

10.2 Program Data

```
F07FSF Example Program Data
  4  2                                                     :Values of N and NRHS
  'L'                                                      :Value of UPLO
 (3.23, 0.00)
 (1.51, 1.92) ( 3.58, 0.00)
 (1.90,-0.84) (-0.23,-1.11) ( 4.09, 0.00)
 (0.42,-2.50) (-1.18,-1.37) ( 2.33, 0.14) ( 4.29, 0.00)  :End of matrix A
 ( 3.93, -6.14) ( 1.48,  6.58)
 ( 6.17,  9.42) ( 4.65, -4.75)
 (-7.17,-21.83) (-4.91,  2.29)
 ( 1.99,-14.38) ( 7.64,-10.79)                             :End of matrix B
```

10.3 Program Results

```
 F07FSF Example Program Results

 Solution(s)
                    1                 2
 1  ( 1.0000,-1.0000) (-1.0000, 2.0000)
 2  (-0.0000, 3.0000) ( 3.0000,-4.0000)
 3  (-4.0000,-5.0000) (-2.0000, 3.0000)
 4  ( 2.0000, 1.0000) ( 4.0000,-5.0000)
```